

# NAG Fortran Library Routine Document

## F07QSF (CSPTRS/ZSPTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F07QSF (CSPTRS/ZSPTRS) solves a complex symmetric system of linear equations with multiple right-hand sides,  $AX = B$ , where  $A$  has been factorized by F07QRF (CSPTRF/ZSPTRF), using packed storage.

### 2 Specification

```
SUBROUTINE F07QSF(UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
ENTRY      csptrs (UPLO, N, NRHS, AP, IPIV, B, LDB, INFO)
INTEGER    N, NRHS, IPIV(*), LDB, INFO
complex   AP(*), B(LDB,*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

To solve a complex symmetric system of linear equations  $AX = B$ , this routine must be preceded by a call to F07QRF (CSPTRF/ZSPTRF) which computes the Bunch–Kaufman factorization of  $A$  using packed storage.

If  $UPLO = 'U'$ ,  $A = PUDU^T P^T$ , where  $P$  is a permutation matrix,  $U$  is an upper triangular matrix and  $D$  is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution  $X$  is computed by solving  $PUDY = B$  and then  $U^T P^T X = Y$ .

If  $UPLO = 'L'$ ,  $A = PLDL^T P^T$ , where  $L$  is a lower triangular matrix; the solution  $X$  is computed by solving  $PLDY = B$  and then  $L^T P^T X = Y$ .

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates how  $A$  has been factorized as follows:

if  $UPLO = 'U'$ ,  $A = PUDU^T P^T$ , where  $U$  is upper triangular;

if  $UPLO = 'L'$ ,  $A = PLDL^T P^T$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides.  
*Constraint:*  $\text{NRHS} \geq 0$ .
- 4: AP(\*) – **complex** array *Input*  
**Note:** the dimension of the array AP must be at least  $\max(1, N * (N + 1)/2)$ .  
*On entry:* details of the factorization of  $A$  stored in packed form, as returned by F07QRF (CSPTRF/ZSPTRF).
- 5: IPIV(\*) – INTEGER array *Input*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On entry:* details of the interchanges and the block structure of  $D$ , as returned by F07QRF (CSPTRF/ZSPTRF).
- 6: B(LDB,\*) – **complex** array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .  
*On exit:* the  $n$  by  $r$  solution matrix  $X$ .
- 7: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F07QSF (CSPTRS/ZSPTRS) is called.  
*Constraint:*  $\text{LDB} \geq \max(1, N)$ .
- 8: INFO – INTEGER *Output*  
*On exit:*  $\text{INFO} = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If  $\text{INFO} = -i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

For each right-hand side vector  $b$ , the computed solution  $x$  is the exact solution of a perturbed system of equations  $(A + E)x = b$ , where

$$|E| \leq c(n)\epsilon P|U||D||U^T|P^T, \text{ if UPLO} = \text{'U'},$$

$$|E| \leq c(n)\epsilon P|L||D||L^T|P^T, \text{ if UPLO} = \text{'L'},$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the **machine precision**.

If  $\hat{x}$  is the true solution, then the computed solution  $x$  satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_\infty}{\|x\|_\infty} \leq c(n) \text{cond}(A, x)\epsilon$$

where  $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_\infty / \|x\|_\infty \leq \text{cond}(A) = \| |A^{-1}| |A| \|_\infty \leq \kappa_\infty(A)$ . Note that  $\text{cond}(A, x)$  can be much smaller than  $\text{cond}(A)$ .

Forward and backward error bounds can be computed by calling F07QVF (CSPRFS/ZSPRFS), and an estimate for  $\kappa_{\infty}(A)$  ( $= \kappa_1(A)$ ) can be obtained by calling F07QUF (CSPCON/ZSPCON).

## 8 Further Comments

The total number of real floating-point operations is approximately  $8n^2r$ .

This routine may be followed by a call to F07QVF (CSPRFS/ZSPRFS) to refine the solution and return an error estimate.

The real analogue of this routine is F07PEF (SSPTRS/DSPTRS).

## 9 Example

To solve the system of equations  $AX = B$ , where

$$A = \begin{pmatrix} -0.39 - 0.71i & 5.14 - 0.64i & -7.86 - 2.96i & 3.80 + 0.92i \\ 5.14 - 0.64i & 8.86 + 1.81i & -3.52 + 0.58i & 5.32 - 1.59i \\ -7.86 - 2.96i & -3.52 + 0.58i & -2.83 - 0.03i & -1.54 - 2.86i \\ 3.80 + 0.92i & 5.32 - 1.59i & -1.54 - 2.86i & -0.56 + 0.12i \end{pmatrix}$$

and

$$B = \begin{pmatrix} -55.64 + 41.22i & -19.09 - 35.97i \\ -48.18 + 66.00i & -12.08 - 27.02i \\ -0.49 - 1.47i & 6.95 + 20.49i \\ -6.43 + 19.24i & -4.59 - 35.53i \end{pmatrix}.$$

Here  $A$  is symmetric, stored in packed form, and must first be factorized by F07QRF (CSPTRF/ZSPTRF).

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07QSF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHMAX, LDB
      PARAMETER        (NMAX=8,NRHMAX=NMAX,LDB=NMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N, NRHS
      CHARACTER        UPLO
*      .. Local Arrays ..
      complex          AP(NMAX*(NMAX+1)/2), B(LDB,NRHMAX)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL         csptrf, csptrs, X04DBF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07QSF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX) THEN
*
*      Read A and B from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
      END IF
```

```

      READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Factorize A
*
      CALL csptrf(UPLO,N,AP,IPIV,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*          Compute solution
*
          CALL csptrs(UPLO,N,NRHS,AP,IPIV,B,LDB,INFO)
*
*          Print solution
*
          IFAIL = 0
          CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed','F7.4',
+                  'Solution(s)','Integer',RLABS,'Integer',CLABS,
+                  80,0,IFAIL)
          ELSE
              WRITE (NOUT,*) 'The factor D is singular'
          END IF
      END IF
      STOP
*
      END

```

## 9.2 Program Data

F07QSF Example Program Data

```

      4  2                                     :Values of N and NRHS
      'L'                                     :Value of UPLO
      (-0.39,-0.71)
      ( 5.14,-0.64) ( 8.86, 1.81)
      (-7.86,-2.96) (-3.52, 0.58) (-2.83,-0.03)
      ( 3.80, 0.92) ( 5.32,-1.59) (-1.54,-2.86) (-0.56, 0.12) :End of matrix A
      (-55.64, 41.22) (-19.09,-35.97)
      (-48.18, 66.00) (-12.08,-27.02)
      ( -0.49, -1.47) ( 6.95, 20.49)
      ( -6.43, 19.24) ( -4.59,-35.53)                                     :End of matrix B

```

## 9.3 Program Results

F07QSF Example Program Results

```

Solution(s)
              1              2
1  ( 1.0000,-1.0000) (-2.0000,-1.0000)
2  (-2.0000, 5.0000) ( 1.0000,-3.0000)
3  ( 3.0000,-2.0000) ( 3.0000, 2.0000)
4  (-4.0000, 3.0000) (-1.0000, 1.0000)

```

---